

Declarative Pearl: Deriving Monadic Quicksort

Shin-Cheng Mu¹ and Tsung-Ju Chiang²

¹ Academia Sinica, Taiwan

² National Taiwan University, Taiwan

Abstract. To demonstrate derivation of monadic programs, we present a specification of sorting using the non-determinism monad, and derive pure quicksort on lists and state-monadic quicksort on arrays. In the derivation one may switch between point-free and pointwise styles, and deploy techniques familiar to functional programmers such as pattern matching and induction on structures or on sizes. Derivation of stateful programs resembles reasoning backwards from the postcondition.

Keywords: monads · program derivation · equational reasoning · non-determinism · state · quicksort

1 Introduction

This pearl presents two derivations of quicksort. The purpose is to demonstrate reasoning and derivation of monadic programs. In the first derivation we present a specification of sorting using the non-determinism monad, from which we derive a pure function that sorts a list. In the second derivation we derive an imperative algorithm, expressed in terms of the state monad, that sorts an array.

Before we dive into the derivations, we shall explain our motivation. Program derivation is the technique of formally constructing a program from a problem specification. In functional derivation, the specification is a function that obviously matches the problem description, albeit inefficiently. It is then stepwise transformed to a program that is efficient enough, where every step is justified by mathematical properties guaranteeing that the program *equals* the specification, that is, for all inputs they compute exactly the same output.

It often happens, for certain problem, that several answers are equally preferred. In sorting, for example, the array to be sorted might contain items with identical keys. It would be inflexible, if not impossible, to decide in the specification how to resolve the tie: it is hard to predict how quicksort arranges items with identical keys before actually deriving quicksort.³ Such problems are better modelled as non-deterministic mappings from the input to all valid outputs. The derived program no longer equals but *refines* the specification.⁴

³ Unless we confine ourselves to stable sorting.

⁴ This is standard in imperative program derivation — Dijkstra [6] argued that we should take non-determinism as default and determinism as a special case.

To cope with non-determinism, there was a trend in the 90's generalising from functions to relations [1,4]. Although these relational calculi are, for advocates including the authors of this paper, concise and elegant, for those who were not following this line of development, these calculi are hard to comprehend and use. People, in their first and often only exposure to the calculi, often complained that the notations are too bizarre, and reasoning with inequality (refinement) too complex. One source of difficulties is that notations of relational calculus are usually *point-free* — that is, about composing relations instead of applying relations to arguments. There have been attempts (e.g [13,5]) designing *pointwise* notations, which functional programmers are more familiar with. Proposals along this line tend to exhibit confusion when functions are applied to non-deterministic values — β -reduction and η -conversion do not hold. One example [13] is that $(\lambda x \rightarrow x - x) (0 \sqcap 1)$, where (\sqcap) denotes non-deterministic choice, always yields 0, while $(0 \sqcap 1) - (0 \sqcap 1)$ could be 0, 1, or -1 .

Preceding the development of relations for program derivation, another way to model non-determinism has gained popularity. Monads [12] were introduced into functional programming as a way to rigorously talk about side effects including IO, state, exception, and non-determinism. Although they are considered one of the main obstacles in learning functional programming (in particular Haskell), monads have gained wide acceptance. In this pearl we propose a calculus of program derivation based on monads — essentially moving to a Kleisli category. Problem specifications are given as Kleisli arrows for non-deterministic monads, to be refined to deterministic functional programs through calculation. One of the benefits is that functional programmers may deploy techniques they are familiar with when reasoning about and deriving programs. These include both point-free and pointwise reasoning, and induction on structures or sizes of data types. An additional benefit of using monads is that we may talk about effects other than non-determinism. We demonstrate how to, from a specification of quicksort on lists, construct the imperative quicksort for arrays. All the derivations and theorems in this pearl are verified in the dependently typed programming language Agda.⁵

2 Monads

A monad consists of a type constructor $m :: * \rightarrow *$ paired with two operators, can be modelled in Haskell as a type class:

VMonad	class Monad <i>m</i> where
vlift	$\{\cdot\} :: a \rightarrow m\ a$
vbind	$(\gg=) :: m\ a \rightarrow (a \rightarrow m\ b) \rightarrow m\ b$.

The operator $\{\cdot\}$ is usually called *return* or *unit*. Since it is used pervasively in this pearl, we use a shorter notation for brevity. One can either think of it

⁵ <https://scm.iis.sinica.edu.tw/home/2020/deriving-monadic-quicksort/>

as mimicking the notation for a singleton set, or C-style syntax for a block of effectful program. They should satisfy the following *monad laws*:

$$\begin{aligned} m \gg \{\cdot\} &= m \quad , \\ \{x\} \gg f &= f \ x \quad , \\ (m \gg f) \gg g &= m \gg (\lambda x \rightarrow f \ x \gg g) \quad . \end{aligned}$$

A standard operator $(\gg) :: \text{Monad } m \Rightarrow m \ a \rightarrow m \ b \rightarrow m \ b$, defined by $m_1 \gg m_2 = m_1 \gg \lambda _ \rightarrow m_2$, is handy when we do not need the result of m_1 . Monadic functions can be combined by Kleisli composition (\gg) , defined by $f \gg g = \lambda x \rightarrow f \ x \gg g$.

Monads usually come with additional operators corresponding to the effects they provide. Regarding non-determinism, we assume two operators \emptyset and (\parallel) , respectively denoting failure and non-deterministic choice:

VMonadPlus	class Monad $m \Rightarrow \text{MonadPlus } m$ where $\emptyset :: m \ a$ $(\parallel) :: m \ a \rightarrow m \ a \rightarrow m \ a \quad .$
vmpepsilon	
vmpadd	

It might be a good time to note that this pearl uses type classes for two purposes: firstly, to be explicit about the effects a program uses. Secondly, the notation implies that it does not matter which actual implementation we use for m , as long as it satisfies all the properties we demand — as Gibbons and Hinze [7] proposed, we use the properties, not the implementations, when reasoning about programs. The style of reasoning in this pearl is not tied to type classes or Haskell, and we do not strictly follow the particularities of type classes in the current Haskell standard.⁶

It is usually assumed that (\parallel) is associative with \emptyset as its identity:

$$\emptyset \parallel m = m = m \parallel \emptyset \quad , \quad (m_1 \parallel m_2) \parallel m_3 = m_1 \parallel (m_2 \parallel m_3) \quad .$$

For the purpose of this pearl, we also demand that (\parallel) be idempotent and commutative. That is, $m \parallel m = m$ and $m \parallel n = n \parallel m$. Efficient implementations of such monads have been proposed (e.g. [10]). However, we use non-determinism monad only in specification. The derived programs are always deterministic.

The laws below concern interaction between non-determinism and (\gg) :

$$\emptyset \gg f = \emptyset \quad , \tag{1}$$

$$m \gg \emptyset = \emptyset \quad , \tag{2}$$

$$(m_1 \parallel m_2) \gg f = (m_1 \gg f) \parallel (m_2 \gg f) \quad , \tag{3}$$

$$m \gg (\lambda x \rightarrow f_1 \ x \parallel f_2 \ x) = (m \gg f_1) \parallel (m \gg f_2) \quad . \tag{4}$$

Left-zero (1) and left-distributivity (3) are standard — the latter says that (\parallel) is algebraic. When mixed with state, right-zero (2) and right-distributivity (4) imply that each non-deterministic branch has its own copy of the state [14].

⁶ For example, we overlook that a **Monad** must also be **Applicative**, **MonadPlus** be **Alternative**, and that functional dependency is needed in a number of places.

3 Specification

We are now ready to present a monadic specification of sorting. Bird [3] demonstrated how to derive various sorting algorithms from relational specifications. In Section 4 and 5 we show how quicksort can be derived in our monadic calculus.

We assume a type Elm (for “elements”) associated with a total preorder (\leq). To sort a list $xs :: \text{List Elm}$ is to choose, among all permutation of xs , those that are sorted:

slowsort $\text{slowsort} :: \text{MonadPlus } m \Rightarrow \text{List Elm} \rightarrow m (\text{List Elm})$
 $\text{slowsort} = \text{perm} \gg \text{filt sorted} \text{ ,}$

where $\text{perm} :: \text{MonadPlus } m \Rightarrow \text{List } a \rightarrow m (\text{List } a)$ non-deterministically computes a permutation of its input, $\text{sorted} :: \text{List Elm} \rightarrow \text{Bool}$ checks whether a list is sorted, and $\text{filt } p \ x$ returns x if $p \ x$ holds, and fails otherwise:

guardBy $\text{filt} :: \text{MonadPlus } m \Rightarrow (a \rightarrow \text{Bool}) \rightarrow a \rightarrow m \ a$
 $\text{filt } p \ x = \text{guard } (p \ x) \gg \{x\} \text{ .}$

The function $\text{guard } b = \text{if } b \text{ then } \{\} \text{ else } \emptyset$ is standard. The predicate $\text{sorted} :: \text{List Elm} \rightarrow \text{Bool}$ can be defined by:

$\text{sorted } [] = \text{True}$
 $\text{sorted } (x : xs) = \text{all } (x \leq) \ xs \wedge \text{sorted } xs \text{ .}$

The following property can be proved by a routine induction on ys :

$$\begin{aligned} \text{sorted } (ys \mathbin{++} [x] \mathbin{++} zs) &\equiv \\ \text{sorted } ys \wedge \text{sorted } zs \wedge \text{all } (\leq x) \ ys \wedge \text{all } (x \leq) \ zs \text{ .} \end{aligned} \quad (5)$$

Now we consider the permutation phase. As shown by Bird [3], what sorting algorithm we end up deriving is often driven by how the permutation phase is performed. The following definition of perm , for example:

$\text{perm } [] = \{\{\}\}$
 $\text{perm } (x : xs) = \text{perm } xs \gg \text{insert } x \text{ ,}$

where $\text{insert } x \ xs$ non-deterministically inserts x into xs , would lead us to insertion sort. To derive quicksort, we use an alternative definition of perm :

$\text{perm} :: \text{MonadPlus } m \Rightarrow \text{List } a \rightarrow m (\text{List } a)$
 $\text{perm } [] = \{\{\}\}$
 $\text{perm } (x : xs) = \text{split } xs \gg \lambda(ys, zs) \rightarrow \text{liftM2 } (\mathbin{++} [x] \mathbin{++}) (\text{perm } ys) (\text{perm } zs) \text{ .}$

where $\text{liftM2 } (\oplus) \ m_1 \ m_2 = m_1 \gg \lambda x_1 \rightarrow m_2 \gg \lambda x_2 \rightarrow \{x_1 \oplus x_2\}$, and split non-deterministically splits a list. When the input has more than one element, we split the tail into two, permute them separately, and insert the head in the middle. The monadic function split is given by:

split $split :: \text{MonadPlus } m \Rightarrow \text{List } a \rightarrow m (\text{List } a \times \text{List } a)$
 $split [] = \{([], [])\}$
 $split (x : xs) = split xs \gg \lambda(ys, zs) \rightarrow \{(x : ys, zs)\} \sqcup \{(ys, x : zs)\} .$

This completes the specification. One may argue that the second definition of *perm* is not one that, as stated in Section 1, “obviously” implied by the problem description. Bird [3] derived the second one from the first in a relational setting, and we can also show that the two definitions are equivalent.

4 Quicksort on Lists

In this section we derive a divide-and-conquer property of *slowsort*. It allows us to refine *slowsort* to the well-known recursive definition of quicksort on lists, and is also used in the next section to construct quicksort on arrays.

Refinement We will need to first define our concept of program refinement. We abuse notations from set theory and define:

RefinesPlus $m_1 \subseteq m_2 \equiv m_1 \sqcup m_2 = m_2 .$

The righthand side $m_1 \sqcup m_2 = m_2$ says that every result of m_1 is a possible result of m_2 . When $m_1 \subseteq m_2$, we say that m_1 *refines* m_1 , m_2 can be *refined to* m_1 , or that m_2 *subsumes* m_1 . Note that this definition applies not only to the non-determinism monad, but to monads having other effects as well. We denote (\subseteq) lifted to functions by $(\dot{\subseteq})$:

RefinesPlusF $f \dot{\subseteq} g = (\forall x : f \ x \subseteq g \ x) .$

That is, f refines g if $f \ x$ refines $g \ x$ for all x . When we use this notation, f and g are always functions returning monads, which is sufficient for this pearl.

One can show that the definition of $(\dot{\subseteq})$ is equivalent to $m_1 \subseteq m_2 \equiv (\exists n : m_1 \sqcup n = m_2)$, and that (\subseteq) and $(\dot{\subseteq})$ are both reflexive, transitive, and anti-symmetric ($m \subseteq n \wedge n \subseteq m \equiv n = m$). Furthermore, (\gg) respects refinement:

Lemma 1. *Bind (\gg) is monotonic with respect to (\subseteq) . That is, $m_1 \subseteq m_2 \Rightarrow m_1 \gg f \subseteq m_2 \gg f$, and $f_1 \dot{\subseteq} f_2 \Rightarrow m \gg f_1 \subseteq m \gg f_2$.*

Having Lemma 1 allows us to refine programs in a compositional manner. The proof of Lemma 1 makes use of (3) and (4).

Commutativity and guard We say that m and n commute if

$$m \gg \lambda x \rightarrow n \gg \lambda y \rightarrow f \ x \ y = n \gg \lambda y \rightarrow m \gg \lambda x \rightarrow f \ x \ y .$$

It can be proved that *guard* p commutes with all m if non-determinism is the only effect in m — a property we will need many times. Furthermore, having

use inductive predicate to qualify this for lemma

right-zero (2) and right-distributivity (4), in addition to other laws, one can prove that non-determinism commutes with other effects. In particular, non-determinism commutes with state.

We mention two more properties about *guard*: *guard* ($p \wedge q$) can be split into two, and *guards* with complementary predicates can be refined to **if**:

guard_disjoint

$$\begin{aligned} \text{guard } (p \wedge q) &= \text{guard } p \gg \text{guard } q \quad , \\ (\text{guard } p \gg m_1) \parallel (\text{guard } (\neg \cdot p) \gg m_2) &\supseteq \text{if } p \text{ then } m_1 \text{ else } m_2 \quad . \end{aligned} \quad (6)$$

Divide-and-Conquer Back to *slowsort*. We proceed with usual routine in functional programming: case-analysis on the input. For the base case, *slowsort* [] = {}. For the inductive case, the crucial step is the commutativity of *guard*:

divide_and_conquer_lemma1

$$\begin{aligned} &\text{slowsort } (p : xs) \\ &= \{ \text{expanding definitions, monad laws} \} \\ &\quad \text{split } xs \gg \lambda(ys, zs) \rightarrow \\ &\quad \text{perm } ys \gg \lambda ys' \rightarrow \text{perm } zs \gg \lambda zs' \rightarrow \\ &\quad \text{filt sorted } (ys' \uplus [p] \uplus zs') \\ &= \{ \text{by (5)} \} \\ &\quad \text{split } xs \gg \lambda(ys, zs) \rightarrow \\ &\quad \text{perm } ys \gg \lambda ys' \rightarrow \text{perm } zs \gg \lambda zs' \rightarrow \\ &\quad \text{guard } (\text{sorted } ys' \wedge \text{sorted } zs' \wedge \text{all } (\leq p) \text{ } ys' \wedge \text{all } (p \leq) \text{ } zs') \gg \\ &\quad \{ys' \uplus [p] \uplus zs'\} \\ &= \{ (6) \text{ and that } \text{guard} \text{ commutes with non-determinism} \} \\ &\quad \text{split } xs \gg \lambda(ys, zs) \rightarrow \text{guard } (\text{all } (\leq p) \text{ } ys \wedge \text{all } (p \leq) \text{ } zs') \gg \\ &\quad (\text{perm } ys \gg \text{filt sorted}) \gg \lambda ys' \rightarrow \\ &\quad (\text{perm } zs \gg \text{filt sorted}) \gg \lambda zs' \rightarrow \\ &\quad \{ys' \uplus [p] \uplus zs'\} \quad . \end{aligned}$$

Provided that we can construct a function *partition* such that

divide_and_conquer_lemma2

$$\{\text{partition } p \text{ } xs\} \subseteq \text{split } xs \gg \text{filt } (\lambda(ys, zs) \rightarrow \text{all } (\leq p) \text{ } ys \wedge \text{all } (p \leq) \text{ } zs) \quad ,$$

we have established the following divide-and-conquer property:

divide_and_conquer

$$\begin{aligned} \text{slowsort } (p : xs) &\supseteq \{\text{partition } p \text{ } xs\} \gg \lambda(ys, zs) \rightarrow \\ &\quad \text{slowsort } ys \gg \lambda ys' \rightarrow \text{slowsort } zs \gg \lambda zs' \rightarrow \\ &\quad \{ys' \uplus [p] \uplus zs'\} \quad . \end{aligned} \quad (8)$$

The derivation of *partition* proceeds by induction on the input. In the case for $xs := x : xs$ we need to refine two guarded choices, (*guard* ($x \leq p$) $\gg \{x : ys, zs\}$) \parallel (*guard* ($p \leq x$) $\gg \{ys, x : zs\}$), to an **if** branching. When x and p equal, the specification allows us to place x in either partition. For no particular reason, we choose the left partition. That gives us:

$$\begin{aligned} \text{partition } p \text{ } [] &= ([], []) \\ \text{partition } p \text{ } (x : xs) &= \text{let } (ys, zs) = \text{partition } p \text{ } xs \\ &\quad \text{in if } x \leq p \text{ then } (x : ys, zs) \text{ else } (ys, x : zs) \quad . \end{aligned}$$

Having *partition* derived, it takes only a routine induction on the length of input lists to show that $\{\cdot\} \cdot \text{qsort} \subseteq \text{slowsort}$, where *qsort* is given by:

$$\begin{aligned} \text{qsort } [] &= [] \\ \text{qsort } (p : xs) &= \text{let } (ys, zs) = \text{partition } p \text{ } xs \\ &\quad \text{in } \text{qsort } ys \mathbin{++} [p] \mathbin{++} \text{qsort } zs \end{aligned}$$

As is typical in program derivation, the termination of derived program is shown separately afterwards. In this case, *qsort* terminates because the input list decreases in size in every recursive call — for that we need to show that, in the call to *partition*, the sum of lengths of *ys* and *zs* equals that of *xs*.

5 Quicksort on Arrays

One of the advantages of using a monadic calculus is that we can integrate effects other than non-determinism into the program we derive. In this section we derive an imperative quicksort on arrays, based on previously established properties.

5.1 Operations on Arrays

We assume that our state is an *Int*-indexed, unbounded array containing elements of type *e*, with two operations that, given an index, respectively read from and write to the array:

```
class Monad m => MonadArr e m where
  read  :: Int -> m e
  write :: Int -> e -> m ()
```

They are assumed to satisfy the following laws:

read-write:	$\text{read } i \gg \text{write } i = \{()\}$,
write-read:	$\text{write } i \ x \gg \text{read } i = \text{write } i \ x \gg \{x\}$,
write-write:	$\text{write } i \ x \gg \text{write } i \ x' = \text{write } i \ x'$,
read-read:	$\text{read } i \gg \lambda x \rightarrow \text{read } i \gg \lambda x' \rightarrow f \ x \ x' =$ $\text{read } i \gg \lambda x \rightarrow f \ x \ x$.

Furthermore, we assume that (1) *read i* and *read j* commute; (2) *write i x* and *write j y* commute if $i \neq j$; (3) *write i x* and *read j* commute if $i \neq j$.

More operations defined in terms of *read* and *write* are shown in Figure 1, where $\#xs$ abbreviates *length xs*. The function *readList i n*, where *n* is a natural number, returns a list containing the *n* elements in the array starting from index *i*. Conversely, *writeList i xs* writes the list *xs* to the array with the first element being at index *i*. In imperative programming we often store sequences of data into an array and return the length of the data. Thus, functions *writeL*, *write2L* and *write3L* store lists into the array before returning their lengths. These *read*

```

readList :: MonadArr e m => Int -> Nat -> m (List e)
readList i 0 = {[]}
readList i (1 + k) = liftM2 (:) (read i) (readList (i + 1) k) ,
writeList :: MonadArr e m => Int -> List e -> m ()
writeList i [] = {()}
writeList i (x : xs) = write i x >> writeList (i + 1) xs ,
writeL i xs = writeList i xs >> {#xs} ,
write2L i (xs, ys) = writeList i (xs ++ ys) >> {(#xs, #ys)} ,
write3L i (xs, ys, zs) = writeList i (xs ++ ys ++ zs) >> {(#xs, #ys, #zs)} .
swap i j = read i >>= λx → read j >>= λy → write i y >> write j x .

```

Fig. 1: Operations for reading and writing chunks of data.

and *write* family of functions are used only in the specification; the algorithm we construct should only mutate the array by *swapping* elements.

Among the many properties of *readList* and *writeList* that can be induced from their definitions, the following will be used in a number of crucial steps:

$$\text{writeList } i (xs ++ ys) = \text{writeList } i xs \gg \text{writeList } (i + \#xs) ys . \quad (9)$$

A function $f :: \text{List } a \rightarrow m (\text{List } a)$ is said to be *length preserving* if $f xs \gg \lambda ys \rightarrow \{(ys, \#ys)\} = f xs \gg \lambda ys \rightarrow \{(ys, \#xs)\}$. It can be proved that *perm*, and thus *slowsort*, are length preserving.

On “composing monads” In the sections to follow, some readers may have concern seeing *perm*, having class constraint **MonadPlus** *m*, and some other code having constraint **MonadArr** *e m* in the same expression. This is totally fine: mixing two such subterms simply results in an expression having constraint $(\text{MonadPlus } m, \text{MonadArr } e m)$. No *lifting* is necessary.

We use type classes to make it clear that we do not specify what exact monad *perm* is implemented with. It could be one monolithic monad, a monad built from monad transformers [8], or a free monad interpreted by effect handlers [11]. All theorems and derivations about *perm* hold regardless of the actual monad, as long as the monad satisfies all properties we demand.

5.2 Partitioning an Array

While the list-based *partition* is relatively intuitive, partitioning an array *in-place* (that is, using at most $O(1)$ additional space) is known to be a tricky phase of array-based quicksort. Therefore we commence our discussion from deriving in-place array partitioning from the list version. The partition algorithm we end up deriving is known as the *Lomuto scheme* [2], as opposed to Hoare’s [9].

Specification There are two issues to deal with before we present a specification for an imperative, array-based partitioning, based on list-based *partition*.

Firstly, *partition* is not tail-recursive, while many linear-time array algorithms are implemented as a tail-recursive *for-loop*. Thus we apply the standard trick constructing a tail-recursive algorithm by introducing accumulating parameters. Define (we write the input/outputs of *partition* in bold font for clarity):

$$\begin{aligned} \text{partl} &:: \text{Elm} \rightarrow (\text{List Elm} \times \text{List Elm} \times \text{List Elm}) \rightarrow (\text{List Elm} \times \text{List Elm}) \\ \text{partl } p \ (ys, zs, \mathbf{xs}) &= \mathbf{let} \ (\mathbf{us}, \mathbf{vs}) = \text{partition } p \ \mathbf{xs} \\ &\quad \mathbf{in} \ (ys \mathbin{++} \mathbf{us}, zs \mathbin{++} \mathbf{vs}) \ . \end{aligned}$$

In words, *partl* *p* (*ys*, *zs*, *xs*) partitions *xs* into (*us*, *vs*) with respect to pivot *p*, but appends *ys* and *zs* respectively to *us* and *vs*. It is a generalisation of *partition* because *partition* *p* *xs* = *partl* *p* ([], [], *xs*). By routine calculation exploiting associativity of (*++*), we can derive a tail-recursive definition of *partl*:

$$\begin{aligned} \text{partl } p \ (ys, zs, []) &= (ys, zs) \\ \text{partl } p \ (ys, zs, \mathbf{x} : \mathbf{xs}) &= \mathbf{if} \ \mathbf{x} \leq p \ \mathbf{then} \ \text{partl } p \ (ys \mathbin{++} [\mathbf{x}], zs, \mathbf{xs}) \\ &\quad \mathbf{else} \ \text{partl } p \ (ys, zs \mathbin{++} [\mathbf{x}], \mathbf{xs}) \ . \end{aligned}$$

It might aid our understanding if we note that, if we start *partl* with initial value ([], [], *xs*) we have the invariant that *ys* contains elements that are at most *p*, and elements in *zs* are larger than *p*. The calculations below, however, do not rely on this observation.⁷

Our wish is to construct a variant of *partl* that works on arrays. That is, when the array contains *ys* *++* *zs* *++* *xs*, the three inputs to *partl* in a consecutive segment, when the derived program finishes its work we wish to have *ys* *++* *us* *++* *zs* *++* *vs*, the output of *partl*, stored consecutively in the array.

This brings us to the second issue: *partition*, and therefore *partl*, are stable (that is, elements in each partition retain their original order), which is a strong requirement for array-based partitioning. It is costly to mutate *ys* *++* *zs* *++* *xs* into *ys* *++* *us* *++* *zs* *++* *vs*, since it demands that we retain the order of elements in *zs* while inserting elements of *us*. For sorting we do not need such a strong postcondition. It is sufficient, and can be done more efficiently, to mutate *ys* *++* *zs* *++* *xs* into *ys* *++* *us* *++* *ws*, where *ws* is some permutation of *zs* *++* *vs*. It is handy allowing non-determinism: we introduce a *perm* in our specification, indicating that we do not care about the order of elements in *ws*.

Define *second* :: Monad *m* ⇒ (*b* → *m* *c*) → (*a*, *b*) → *m* (*a*, *c*), which applies a monadic function to the second component of a tuple:

$$\text{second } f \ (x, y) = f \ y \gg= \lambda y' \rightarrow \{(x, y')\} \ .$$

Our new wish is to construct an array counterpart of *second perm · partl p*. Let the function be

$$\begin{aligned} \text{ipartl} &:: (\text{MonadPlus } m, \text{MonadArr Elm } m) \Rightarrow \\ &\quad \text{Elm} \rightarrow \text{Int} \rightarrow (\text{Nat} \times \text{Nat} \times \text{Nat}) \rightarrow m \ (\text{Nat} \times \text{Nat}) \ . \end{aligned}$$

⁷ It might be worth noting that *partl* causes a space leak in Haskell, since the accumulators become thunks that increase in size as the input list is traversed. It does not matter here since *partl* merely serves as a specification of *ipartl*.

The intention is that in a call $ipartl\ p\ i\ (ny, nz, nx)$, p is the pivot, i the index where $ys \uparrow zs \uparrow xs$ is stored in the array, and ny, nz, nx respectively the lengths of ys , zs , and xs . A specification of $ipartl$ is:

$$writeList\ i\ (ys \uparrow zs \uparrow xs) \gg ipartl\ p\ i\ (\#ys, \#zs, \#xs) \subseteq \\ second\ perm\ (partl\ p\ (ys, zs, xs)) \gg write2L\ i\ .$$

That is, under assumption that $ys \uparrow zs \uparrow xs$ is stored in the array starting from index i (initialised by $writeList$), $ipartl$ computes $partl\ p\ (ys, zs, xs)$, possibly permuting the second partition. The resulting two partitions are still stored in the array starting from i , and their lengths are returned.

Derivation We start with fusing $second\ perm$ into $partl$, that is, to construct $partl'\ p \subseteq second\ perm \cdot partl\ p$.⁸ If we discover an inductive definition of $partl'$, it can then be used to construct an inductive definition of $ipartl$. With some routine calculation we get:

$$partl' :: MonadPlus\ m \Rightarrow Elm \rightarrow (List\ Elm)^3 \rightarrow m\ (List\ Elm \times List\ Elm) \\ partl'\ p\ (ys, zs, []) = \{(ys, zs)\} \\ partl'\ p\ (ys, zs, x : xs) = \\ \quad \text{if } x \leq p \text{ then } perm\ zs \gg \lambda zs' \rightarrow partl'\ p\ (ys \uparrow [x], zs', xs) \\ \quad \text{else } perm\ (zs \uparrow [x]) \gg \lambda zs' \rightarrow partl'\ p\ (ys, zs', xs) \ .$$

For an intuitive explanation, rather than permuting the second list zs after computing $partl$, we can also permute zs in $partl'$ before every recursive call.

The specification of $ipartl$ now becomes

ipartl_spec

$$writeList\ i\ (ys \uparrow zs \uparrow xs) \gg ipartl\ p\ i\ (\#ys, \#zs, \#xs) \subseteq \\ partl'\ p\ (ys, zs, xs) \gg write2L\ i\ . \quad (10)$$

ultimately trying to find
ipartl that satisfies this, using
this def of partl'

To calculate $ipartl$, we start with the right-hand side of (\subseteq), since it contains more information to work with. We try to push $write2L$ leftwards until the expression has the form $writeList\ i\ (ys \uparrow zs \uparrow xs) \gg \dots$, thereby constructing $ipartl$. This is similar to that, in imperative program calculation, we *work backwards from the postcondition* to construct a program that works under the given precondition [6].

We intend to construct $ipartl$ by induction on xs . For $xs := []$, we get $ipartl\ p\ i\ (ny, nz, 0) = \{(ny, nz)\}$. For the case $x : xs$, assume that the specification is met for xs . Just for making the calculation shorter, we refactor $partl'$, lifting the recursive calls and turning the main body into an auxiliary function:

$$partl'\ p\ (ys, zs, x : xs) = dispatch\ x\ p\ (ys, zs, xs) \gg partl'\ p\ , \\ \text{where } dispatch\ x\ p\ (ys, zs, xs) = \\ \quad \text{if } x \leq p \text{ then } perm\ zs \gg \lambda zs' \rightarrow \{(ys \uparrow [x], zs', xs)\} \\ \quad \text{else } perm\ (zs \uparrow [x]) \gg \lambda zs' \rightarrow \{(ys, zs', xs)\} \ .$$

⁸ We will discover a stronger specification $partl'\ p \subseteq snd3\ perm \setminus (second\ perm \cdot partl\ p)$, where $snd3\ f\ (x, y, z) = f\ y \gg \lambda y' \rightarrow \{(x, y', z)\}$. We omit the details.

We calculate:

$$\begin{aligned}
& \text{partl}' p (ys, zs, \mathbf{x} : \mathbf{xs}) \gg \text{write2L } i \\
= & \{ \text{definition of } \text{partl}' \} \\
& (\text{dispatch } \mathbf{x} p (ys, zs, \mathbf{xs}) \gg \text{partl}' p) \gg \text{write2L } i \\
\supseteq & \{ \text{monad laws, inductive assumption} \} \\
& (\text{dispatch } \mathbf{x} p (ys, zs, \mathbf{xs}) \gg \text{write3L } i) \gg \text{ipartl } p i \\
= & \{ \text{by (9), monad laws} \} \\
& \text{dispatch } \mathbf{x} p (ys, zs, \mathbf{xs}) \gg \lambda(ys', zs', \mathbf{xs}) \rightarrow \\
& \text{writeList } i (ys' + zs') \gg \text{writeList } (i + \#(ys' + zs')) \mathbf{xs} \gg \\
& \text{ipartl } p i (\#ys', \#zs', \# \mathbf{xs}) \\
= & \{ \text{perm preserves length, commutativity} \} \\
& \text{writeList } (i + \#ys + \#zs + 1) \mathbf{xs} \gg \\
& \text{dispatch } \mathbf{x} p (ys, zs, \mathbf{xs}) \gg \lambda(ys', zs', \mathbf{xs}) \rightarrow \\
& \text{writeList } i (ys' + zs') \gg \\
& \text{ipartl } p i (\#ys', \#zs', \# \mathbf{xs}) \\
= & \{ \text{definition of } \text{dispatch}, \text{ function calls distribute into } \mathbf{if} \} \\
& \text{writeList } (i + \#ys + \#zs + 1) \mathbf{xs} \gg \\
& \mathbf{if } \mathbf{x} \leq p \text{ then } \text{perm } zs \gg \lambda zs' \rightarrow \text{writeList } i (ys + [\mathbf{x}] + zs') \gg \\
& \quad \text{ipartl } p i (\#ys + 1, \#zs', \# \mathbf{xs}) \\
& \quad \mathbf{else } \text{perm } (zs + [\mathbf{x}]) \gg \lambda zs' \rightarrow \text{writeList } i (ys + zs') \gg \\
& \quad \text{ipartl } p i (\#ys, \#zs', \# \mathbf{xs}) .
\end{aligned}$$

We pause here to see what has happened: we have constructed a precondition $\text{writeList } (i + \#ys + \#zs + 1) \mathbf{xs}$, which is part of the desired precondition: $\text{writeList } i (ys + zs + (\mathbf{x} : \mathbf{xs}))$. To recover the latter precondition, we will try to turn both branches of **if** into the form $\text{writeList } i (ys + zs + [\mathbf{x}]) \gg \dots$. That is, we try to construct, in both branches, some code that executes under the precondition $\text{writeList } i (ys + zs + [\mathbf{x}])$ — that the code generates the correct result is guaranteed by the refinement relation.

It is easier for the second branch, where we can simply refine perm to $\{\cdot\}$:

$$\begin{aligned}
& \text{perm } (zs + [\mathbf{x}]) \gg \lambda zs' \rightarrow \text{writeList } i (ys + zs') \gg \\
& \text{ipartl } p i (\#ys, \#zs', \# \mathbf{xs}) \\
\supseteq & \{ \text{since } \{xs\} \subseteq \text{perm } xs \} \\
& \text{writeList } i (ys + zs + [\mathbf{x}]) \gg \text{ipartl } p i (\#ys, \#zs + 1, \# \mathbf{xs}) .
\end{aligned}$$

For the first branch, we focus on its first line:

$$\begin{aligned}
& \text{perm } zs \gg \lambda zs' \rightarrow \text{writeList } i (ys + [\mathbf{x}] + zs') \\
= & \{ \text{by (9), commutativity} \} \\
& \text{writeList } i ys \gg \text{perm } zs \gg \lambda zs' \rightarrow \text{writeList } (i + \#ys) ([\mathbf{x}] + zs') \\
\supseteq & \{ \text{introduce } \text{swap}, \text{ see below} \} \\
& \text{writeList } i ys \gg \text{writeList } (i + \#ys) (zs + [\mathbf{x}]) \gg \\
& \text{swap } (i + \#ys) (i + \#ys + \#zs) \\
= & \{ \text{by (9)} \} \\
& \text{writeList } i (ys + zs + [\mathbf{x}]) \gg \text{swap } (i + \#ys) (i + \#ys + \#zs) .
\end{aligned}$$

ipartl_spec_lemma1

ipartl_spec_lemma2

Here we explain the last two steps. Operationally speaking, given an array containing $ys \mathbin{++} zs \mathbin{++} [x]$ (the precondition we wanted, initialized by the *writeList* in the last line), how do we mutate it to $ys \mathbin{++} [x] \mathbin{++} zs'$ (postcondition specified by the *writeList* in the first line), where zs' is a permutation of zs ? We may do so by swapping x with the leftmost element of zs , which is what we did in the second step. Formally, we used the property:

$$\boxed{\text{ipartl_spec_lemma3}} \quad \begin{aligned} \text{perm } zs \gg \lambda zs' \rightarrow \text{writeList } i ([x] \mathbin{++} zs') \supseteq \\ \text{writeList } i (zs \mathbin{++} [x]) \gg \text{swap } i (i + \#zs) \end{aligned} \quad (11)$$

Now that both branches are refined to code with precondition *writeList* $i (ys \mathbin{++} zs \mathbin{++} [x])$, we go back to the main derivation:

$$\begin{aligned} & \text{writeList } (i + \#ys + \#zs + 1) \text{ } \mathbf{x} \mathbf{s} \gg \\ & \text{if } \mathbf{x} \leq p \text{ then } \text{writeList } i (ys \mathbin{++} zs \mathbin{++} [x]) \gg \\ & \quad \text{swap } (i + \#ys) (i + \#ys + \#zs) \gg \\ & \quad \text{ipartl } p \ i \ (\#ys + 1, \#zs, \# \mathbf{x} \mathbf{s}) \\ & \quad \text{else } \text{writeList } i (ys \mathbin{++} zs \mathbin{++} [x]) \gg \\ & \quad \text{ipartl } p \ i \ (\#ys, \#zs + 1, \# \mathbf{x} \mathbf{s}) \\ = & \quad \{ \text{distributivity of if, (9)} \} \\ & \text{writeList } i (ys \mathbin{++} zs \mathbin{++} (\mathbf{x} : \mathbf{x} \mathbf{s})) \gg \\ & \text{if } \mathbf{x} \leq p \text{ then } \text{swap } (i + \#ys) (i + \#ys + \#zs) \gg \\ & \quad \text{ipartl } p \ i \ (\#ys + 1, \#zs, \# \mathbf{x} \mathbf{s}) \\ & \quad \text{else } \text{ipartl } p \ i \ (\#ys, \#zs + 1, \# \mathbf{x} \mathbf{s}) \\ = & \quad \{ \text{write-read and definition of writeList} \} \\ & \text{writeList } i (ys \mathbin{++} zs \mathbin{++} (\mathbf{x} : \mathbf{x} \mathbf{s})) \gg \\ & \text{read } (i + \#ys + \#zs) \gg \lambda \mathbf{x} \rightarrow \\ & \text{if } \mathbf{x} \leq p \text{ then } \text{swap } (i + \#ys) (i + \#ys + \#zs) \gg \\ & \quad \text{ipartl } p \ i \ (\#ys + 1, \#zs, \# \mathbf{x} \mathbf{s}) \\ & \quad \text{else } \text{ipartl } p \ i \ (\#ys, \#zs + 1, \# \mathbf{x} \mathbf{s}) \end{aligned}$$

We have thus established the precondition *writeList* $i (ys \mathbin{++} zs \mathbin{++} (\mathbf{x} : \mathbf{x} \mathbf{s}))$. In summary, we have derived:

$$\begin{aligned} \text{ipartl} &:: \text{MonadArr Elm } m \Rightarrow \text{Elm} \rightarrow \text{Int} \rightarrow (\text{Int} \times \text{Int} \times \text{Int}) \rightarrow m (\text{Int} \times \text{Int}) \\ \text{ipartl } p \ i \ (ny, nz, 0) &= \{(ny, nz)\} \\ \text{ipartl } p \ i \ (ny, nz, 1 + k) &= \\ & \text{read } (i + ny + nz) \gg \lambda x \rightarrow \\ & \text{if } x \leq p \text{ then } \text{swap } (i + ny) (i + ny + nz) \gg \text{ipartl } p \ i \ (ny + 1, nz, k) \\ & \text{else } \text{ipartl } p \ i \ (ny, nz + 1, k) \end{aligned}$$

5.3 Sorting an Array

Now that we have *ipartl* derived, the rest of the work is to install it into quicksort. We intend to derive $\text{iqsort} :: \text{MonadArr Elm } m \Rightarrow \text{Int} \rightarrow \text{Nat} \rightarrow m ()$ such that

$isort\ i\ n$ sorts the n elements in the array starting from index i . We can give it a formal specification:

$$\boxed{\text{iqsort_spec}} \quad writeList\ i\ xs \gg iqsort\ i\ (\#xs) \subseteq slowsort\ xs \gg writeList\ i \ . \quad (12)$$

That is, when $iqsort\ i$ is run from a state initialised by $writeList\ i\ xs$, it should behave the same as $slowsort\ xs \gg writeList\ i$.

The function $iqsort$ can be constructed by induction on the length of the input list. For the case $xs := p : xs$, we start from the left-hand side $slowsort\ (p : xs) \gg writeList\ i$ and attempt to transform it to $writeList\ i\ (p : xs) \gg \dots$, thereby construct $iqsort$. We present only the highlights of the derivation. Firstly, $slowsort\ (p : xs) \gg writeList\ i$ can be transformed to:

$$\boxed{\text{iqsort_spec_lemma1}} \quad \begin{aligned} &partl'\ p\ ([], [], xs) \gg \lambda(ys, zs) \rightarrow \\ &perm\ ys \gg \lambda ys' \rightarrow writeList\ i\ (ys' \uparrow [p] \uparrow zs) \gg \\ &iqsort\ i\ (\#ys) \gg iqsort\ (i + \#ys + 1)\ (\#zs) \ . \end{aligned}$$

For that to work, we introduced two $perm$ to permute both partitions generated by $partition$. We can do so because $perm \gg perm = perm$ and thus $perm \gg slowsort = slowsort$. The term $perm\ zs$ was combined with $partition\ p$, yielding $partl'\ p$, while $perm\ ys$ will be needed later. We also needed (9) to split $writeList\ i\ (ys' \uparrow [p] \uparrow zs)$ into two parts. Assuming that (12) has been met for lists shorter than xs , two subexpressions are folded back to $iqsort$.

Now that we have introduced $partl'$, the next goal is to embed $ipartl$. The status of the array before the two calls to $iqsort$ is given by $writeList\ i\ (ys' \uparrow [p] \uparrow zs)$. That is, $ys' \uparrow [p] \uparrow zs$ is stored in the array from index i , where ys' is a permutation of ys . The postcondition of $ipartl$, according to the specification (10), ends up with ys and zs stored consecutively. To connect the two conditions, we use a lemma that is dual to (11):

$$\boxed{\text{iqsort_spec_lemma2}} \quad \begin{aligned} &perm\ ys \gg \lambda ys' \rightarrow writeList\ i\ (ys' \uparrow [p]) \supseteq \\ &writeList\ i\ ([p] \uparrow ys) \gg swap\ i\ (i + \#ys) \ . \end{aligned} \quad (13)$$

This is what the typical quicksort algorithm does: swapping the pivot with the last element of ys , and (13) says that it is valid because that is one of the many permutations of ys . With (13) and (10), the specification can be refined to:

$$\boxed{\text{iqsort_spec_lemma3}} \quad \begin{aligned} &writeList\ i\ (p : xs) \gg \\ &ipartl\ p\ (i + 1)\ (0, 0, \#xs) \gg \lambda(ny, nz) \rightarrow swap\ i\ (i + ny) \gg \\ &iqsort\ i\ (\#ys) \gg iqsort\ (i + \#ys + 1)\ (\#zs) \ . \end{aligned}$$

In summary, we have derived:

$$\begin{aligned} iqsort &:: \text{MonadArr Elm } m \Rightarrow \text{Int} \rightarrow \text{Nat} \rightarrow m () \\ iqsort\ i\ 0 &= \{()\} \\ iqsort\ i\ n &= read\ i \gg \lambda p \rightarrow \\ &\quad ipartl\ p\ (i + 1)\ (0, 0, n - 1) \gg \lambda(ny, nz) \rightarrow \\ &\quad swap\ i\ (i + ny) \gg \\ &\quad iqsort\ i\ ny \gg iqsort\ (i + ny + 1)\ nz \ . \end{aligned}$$

6 Conclusions

From a specification of sorting using the non-determinism monad, we have derived a pure quicksort for lists and a state-monadic quicksort for arrays. We hope to demonstrate that the monadic style is a good choice as a calculus for program derivation that involves non-determinism. One may perform the derivation in pointwise style, and deploy techniques that functional programmers have familiarised themselves with, such as pattern matching and induction on structures or on sizes. When preferred, one can also work in point-free style with (\gg) . Programs having other effects can be naturally incorporated into this framework. The way we derive stateful programs echos how we, in Dijkstra's style, reason backwards from the postcondition.

A final note: (\gg) and $(\dot{\subseteq})$ naturally induce the notion of (left) factor, $(\backslash) :: (a \rightarrow m\ b) \rightarrow (a \rightarrow m\ c) \rightarrow b \rightarrow m\ c$, defined by the Galois connection:

$$f \gg g \dot{\subseteq} h \equiv g \dot{\subseteq} f \backslash h .$$

Let $h :: a \rightarrow m\ c$ be a monadic specification, and $f :: a \rightarrow m\ b$ performs the computation half way, then $f \backslash h$ is the most non-deterministic (least constrained) monadic program that, when ran after the postcondition set up by f , still meets the result specified by h . With (\backslash) , *ipartl* and *iqsort* can be specified by:

$$\begin{aligned} \text{ipartl } p\ i &\dot{\subseteq} \text{write3L } i \backslash ((\text{second perm} \cdot \text{partl } p) \gg \text{write2L } i) , \\ \text{iqsort } i &\dot{\subseteq} \text{writeL } i \backslash (\text{slowsort} \gg \text{writeList } i) . \end{aligned}$$

In relational calculus, the *factor* is an important operator that is often associated with weakest precondition. We unfortunately cannot cover it due to space constraints.

Acknowledgements The authors would like to thank Jeremy Gibbons for the valuable discussions during development of this work.

References

1. Backhouse, R.C., de Bruin, P.J., Malcolm, G., Voermans, E., van der Woude, J.: Relational catamorphisms. In: Möller, B. (ed.) Proceedings of the IFIP TC2/WG2.1 Working Conference on Constructing Programs. pp. 287–318. Elsevier Science Publishers (1991)
2. Bentley, J.L.: Programming Pearls, Second Edition. Addison-Wesley (2000)
3. Bird, R.S.: Functional algorithm design. Science of Computer Programming **26**, 15–31 (1996)
4. Bird, R.S., de Moor, O.: Algebra of Programming. International Series in Computer Science, Prentice Hall (1997)
5. Bird, R.S., Rabe, F.: How to calculate with nondeterministic functions. In: Hutton, G. (ed.) Mathematics of Program Construction. pp. 138–154. Springer (2019)
6. Dijkstra, E.W.: A Discipline of Programming. Prentice Hall (1976)

7. Gibbons, J., Hinze, R.: Just do it: simple monadic equational reasoning. In: Danvy, O. (ed.) International Conference on Functional Programming. pp. 2–14. ACM Press (2011)
8. Gill, A., Kmett, E.: The monad transformer library. <https://hackage.haskell.org/package/mtl> (2014)
9. Hoare, C.A.R.: Algorithm 63: Partition. Communications of the ACM 4(7), 321 (1961)
10. Kiselyov, O.: How to restrict a monad without breaking it: the winding road to the Set monad. <http://okmij.org/ftp/Haskell/set-monad.html> (July 2013)
11. Kiselyov, O., Ishii, H.: Freer monads, more extensible effects. In: Reppy, J.H. (ed.) Symposium on Haskell. pp. 94–105. ACM Press (2015)
12. Moggi, E.: Computational lambda-calculus and monads. In: Parikh, R. (ed.) Logic in Computer Science. pp. 14–23. IEEE Computer Society Press (1989)
13. de Moor, O., Gibbons, J.: Pointwise relational programming. In: Rus, T. (ed.) Algebraic Methodology and Software Technology. pp. 371–390. No. 1816 in Lecture Notes in Computer Science, Springer-Verlag (2000)
14. Pauwels, K., Schrijvers, T., Mu, S.C.: Handling local state with global state. In: Hutton, G. (ed.) Mathematics of Program Construction. pp. 18–44. Springer (2019)